Запрос должен получать реализации, которых нет во ВТ, а выводит 0 результатов.
Может глаза замылились и не вижу очевидной ошибки?
ПениЗаПросрочкуКредита.РеализацияОплатаПени - это измерение "ДокументСсылка.РеализацияТоваровУслуг";

Запрос с ВТ не работает.
```1c
    ВЫБРАТЬ
    ПениЗаПросрочкуКредита.РеализацияОплатаПени КАК РеализацияОплатаПени
    ПОМЕСТИТЬ ВременнаяТаблица
    ИЗ
    РегистрСведений.ПениЗаПросрочкуКредита КАК ПениЗаПросрочкуКредита
    ГДЕ
    ПениЗаПросрочкуКредита.РеализацияОплатаПени <> ЗНАЧЕНИЕ(документ.реализациятоваровуслуг.пустаяссылка)
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    РеализацияТоваровУслугТовары.Сумма КАК ОплаченоПени,
    РеализацияТоваровУслугТовары.Ссылка.Контрагент КАК Контрагент,
    РеализацияТоваровУслугТовары.Ссылка.Договор КАК Договор,
    РеализацияТоваровУслугТовары.Ссылка КАК Реализация
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары,
    ВременнаяТаблица КАК ВременнаяТаблица
    ГДЕ
    РеализацияТоваровУслугТовары.Номенклатура.Код = "НЦ000001841"
    И РеализацияТоваровУслугТовары.Ссылка.Статус = ЗНАЧЕНИЕ(перечисление.СтатусыРеализацийТоваровУслуг.отгружено)
    И НЕ РеализацияТоваровУслугТовары.Ссылка В (ВременнаяТаблица.РеализацияОплатаПени)
```

Проверял без ВТ - работает:
Рабочий запрос без ВТ, через массив.
```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ПениЗаПросрочкуКредита.РеализацияОплатаПени КАК РеализацияОплатаПени
    |ИЗ
    | РегистрСведений.ПениЗаПросрочкуКредита КАК ПениЗаПросрочкуКредита
    |ГДЕ
    | ПениЗаПросрочкуКредита.РеализацияОплатаПени <> ЗНАЧЕНИЕ(документ.реализациятоваровуслуг.пустаяссылка)";

    РезультатЗапроса = Запрос.Выполнить();

    ВыборкаДетальныеЗаписи = РезультатЗапроса.Выбрать();

    массивОплатаПени = Новый Массив;

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл

    массивОплатаПени.Добавить(ВыборкаДетальныеЗаписи.РеализацияОплатаПени);

    КонецЦикла;

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | РеализацияТоваровУслугТовары.Сумма КАК ОплаченоПени,
    | РеализацияТоваровУслугТовары.Ссылка.Контрагент КАК Контрагент,
    | РеализацияТоваровУслугТовары.Ссылка.Договор КАК Договор,
    | РеализацияТоваровУслугТовары.Ссылка КАК Реализация
    |ИЗ
    | Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    |ГДЕ
    | РеализацияТоваровУслугТовары.Номенклатура.Код = "НЦ000001841"
    | И РеализацияТоваровУслугТовары.Ссылка.Статус = ЗНАЧЕНИЕ(перечисление.СтатусыРеализацийТоваровУслуг.отгружено)
    | И НЕ РеализацияТоваровУслугТовары.Ссылка В (&массивОплатаПени)";

    Запрос.УстановитьПараметр("массивОплатаПени", массивОплатаПени);
```
---
```1c
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары

    ГДЕ
    РеализацияТоваровУслугТовары.Номенклатура.Код = "НЦ000001841"
    И РеализацияТоваровУслугТовары.Ссылка.Статус = ЗНАЧЕНИЕ(перечисление.СтатусыРеализацийТоваровУслуг.отгружено)
    И НЕ РеализацияТоваровУслугТовары.Ссылка В (
    ВЫБРАТЬ
    ВременнаяТаблица.РеализацияОплатаПени
    ИЗ
    ВременнаяТаблица КАК ВременнаяТаблица
    )
```
---
```1c
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары

    ГДЕ
    РеализацияТоваровУслугТовары.Номенклатура.Код = "НЦ000001841"
    И РеализацияТоваровУслугТовары.Ссылка.Статус = ЗНАЧЕНИЕ(перечисление.СтатусыРеализацийТоваровУслуг.отгружено)
    И НЕ РеализацияТоваровУслугТовары.Ссылка В (
    ВЫБРАТЬ
    ВременнаяТаблица.РеализацияОплатаПени
    ИЗ
    ВременнаяТаблица КАК ВременнаяТаблица
    )
```
---
Спасибо, заработало. Просьба объяснить, почему заработало)

Если ВТ убираем ИЗ выбранных таблиц, то обратится к ВТ через условие не получается... Пишем запрос в самом условии - это как-то не очевидно.
---
Если ВТ убираем ИЗ выбранных таблиц, то обратится к ВТ через условие не получается...
вот это не понял.
Тут просто вложенный запрос в условии. Это нормальная практика обращения к временной таблице.
---
в консоле запросов, конструктор вкладка "Условия" - нет ВТ, если не добавить её в список таблиц. Думал раз нет её там то и обратится к её данным нельзя.
---
ну, это вопросы к консоле. А так это обычный вложенный запрос. Тут можно было выбирать напрямую из физических таблиц. А можно как и выше, из временной таблицы.

И еще, я бы посоветовал еще проиндексировать временную таблицу. Тут может быть принципиально сказаться на быстродействии.
---
а объяснение простое.
ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары,
    ВременнаяТаблица КАК ВременнаяТаблица
Это декартово произведение. Всех элементов из первой таблицы со всеми элементами из второй таблицы.
Просто не выводите ничего из 2 таблицы, вот и не видно. Но под капотом все элементы результат запроса имеют данные из 2 таблицы.
Вот и выводит 0 результата.